
ABSTRACT 

This paper is about using SAS to retrieve financial data from submitted on-line Excel templates. 

INTRODUCTION 

The Texas Education Agency, TEA, began providing an on-line template for charter schools to submit their annual 

financial data to the TEA’s Financial Accountability Division in the 2013-2014 school year when its use was 

voluntary. Using the template will be mandatory in the 2014-2015 school year. The charter school’s financial data 

was previously submitted to the TEA’s Financial Compliance Division and the data was stored in an MS Access 

database which is the audit division’s preferred data storage and analysis tool. The Financial Accountability 

Division, the division I’m in, uses SAS extensively and some examples are: fulfilling data requests, web reporting, 

federal reporting, and data analysis. A SAS solution was needed to retrieve the data from the submitted on-line 

Excel templates for storage in a MS Access database, and this paper is about the SAS program that was developed to 

meet that need. The SAS program uses: macros to reduce the amount of code and make the program easier to 

manage, the “DICTIONARY.COLUMNS” table, proc datasets, the "vlable" function, and proc append. Also, 

knowledge of BASE SAS and SAS macros will make it is easier to understand and apply the concepts presented in 

this paper.  

SAS CODE 

We’re going to go through the SAS code a step at a time, and the complete SAS code is at the end of this paper. The 

SAS code in Table A is the first part of a macro that is called “import_data” and it has the macro variables “tab” (the 

seven spreadsheet tab names) and “ds” (the seven temporary SAS datasets names), and it is inside another macro 

that is called “file_info” and it has the macro variable “file_name” (the spreadsheet filename). The proc import is 

called seven times per spreadsheet and imports the seven tabs of a template into seven temporary SAS datasets.    
 

Table A 

 

 

 

 

 

 

Table B is the proc contents of one of temporary SAS datasets after being imported and please note variable 2. 

 

Table B 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

PROC IMPORT OUT= work.&ds. 

DATAFILE= "S:\fiscal\AFR\Submitted 2013 Charter template AFRs\&file_name."  

DBMS=EXCEL REPLACE; RANGE="&tab"; GETNAMES=YES; MIXED=NO; SCANTEXT=YES; 

USEDATE=YES; SCANTIME=YES; RUN; 

 

# Variable Label 

1 F1 F1 

2 _15_820___SAN_ANTONIO_SCHOOL_FOR 

015-820   SAN ANTONIO SCHOOL FOR INQUIRY & 

CREATIVITY 

3 F3 F3 

4 F4 F4 

5 F5 F5 

6 F6 F6 

7 F7 F7 

8 F8 F8 

9 F9 F9 

10 F10 F10 

11 F11 F11 

The SAS code in Table C renames the variables into a consistent pattern so the County District Number, CDN, can 

be retrieved from the 2
nd

 variable’s label as shown in Table B dynamically. The proc sql gets the max column 

number and makes a list of the variable names from the SAS dataset’s dictionary.columns table. The proc datasets 

renames the variable names into a consistent pattern, F_1-F_max column number, so the vlable function can get the 

CDN from the 2
nd

 variable’s label as shown in the last lines of code in Table C. 

 

Table C 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

Table D is the proc contents after the SAS code in Table C, and please note variable 3 has been renamed from 

“_15_820___SAN_ANTONIO_SCHOOL_FOR” to “F_2” & variable 1 now contains the CDN “015820”. 

 
Table D 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

# Variable Label 

1 CDNum 

 2 F_1 F1 

3 F_2 

015-820   SAN ANTONIO SCHOOL FOR INQUIRY & 

CREATIVITY 

4 F_3 F3 

5 F_4 F4 

6 F_5 F5 

7 F_6 F6 

8 F_7 F7 

9 F_8 F8 

10 F_9 F9 

11 F_10 F10 

12 F_11 F11 

 

title1; 

proc sql; 

select max(varnum) into :num_vars 

FROM DICTIONARY.COLUMNS 

WHERE UPCASE(LIBNAME)="WORK" AND UPCASE(memname)="&ds."; 

select distinct(name) into :var1-:var%TRIM(%LEFT(&num_vars)) 

from dictionary.columns 

WHERE UPCASE(LIBNAME)="WORK" AND UPCASE(memname)="&ds." 

order by varnum; 

quit; 

proc datasets library=WORK; 

modify &ds.; 

rename 

%do i=1 %to &num_vars; 

&&var&i=F_&i. 

%end;  

; 

quit; run; 

 

%if &ds. NE COVER %then %do; data &ds.; length CDNum $6; set &ds.; 

CDNum=compress(substr(vlabel(F_2),1,7),"-"); run; %end; 

The SAS code in Table E makes the temporary SAS datasets that contain the variables that will be loaded into the MS 

Access database. The SAS code is a macro that is called “data” and it has the macro variables “var” (variable name), 

“acct” (the label of the line item in the template, “ds” (temporary SAS dataset), and “col” (the column that contains 

the value of the variable in the template). The %data line in the code is an example of one of the macro calls, and it 

extracts “Local Revenue” from the template. 

 
Table E 

 

 

 

 

 

 

 

The SAS code in Table F merges the data for each template together, and the proc append concatenates the data into a 

SAS dataset that contains the data from all of the templates.  

 
Table F 

 

 

 

 

 

 

 

 

 

The SAS code in Table G exports the data to Excel so the data can be imported into MS Access. Please note the data 

could have also been exported directly from SAS into MS Access but the Financial Compliance Division still owns 

the MS Access database so letting them load the spreadsheet into the database should work well. 

 
Table G 

 

 

 

 

 

 

 

CONCLUSION 

This paper covered using SAS to retrieve financial data from submitted on-line Excel templates so it can be 

imported into a MS Access database. 

%macro data(var,acct,ds,col); 

data work.&var.(keep=CDNum &var.); set work.&ds.(where=(F_2 = &acct.)); &var.=input(&col.,dollar32.); 

if &var.=. then &var.=0; run; 

%mend data; 

 

%data(LR,"Local revenue",Budget_comparison,F_8); 

data work.charter_AFR_&file_number.;  

merge CDNum CA COH COHPR CHCOH OA TA CL OL TL UNA TRNA PRNA TNA INV REC PREPD 

AP NPC NPL DE CUNA URI CHFR CHFE LND PE ADEP LR SR FR TR F11 F12 F13 F21 F23 F31 F32 

F33 F34 F35 F36 F41 F51 F52  

F53 F61 F71 F81 TE OtherGainsLosses PriorPeriodAdjustments CNA O6100 O6200 O6300 O6400 O6500 

O7900 O8900 TOE; 

by CDNum; 

run; 

 

proc append base=work.charter_AFR data=work.charter_AFR_&file_number.; 

run; 

%macro export(data); 

 PROC EXPORT DATA= WORK.&data 

 OUTFILE= "S:\fiscal\AFR\Charter AFR data\&data. &filedate..xlsx" 

 DBMS=EXCEL REPLACE; 

 SHEET="&data."; 

RUN; 

%mend export; 

%export(charter_AFR); 

COMPLETE SAS PROGRAM 

options mlogic symbolgen macrogen pageno=1; 

 

%let filedate = %left(%qsysfunc(date(),mmddyyn8.)); 

%macro file_info(file_name,file_number); 

%macro import_data(tab,ds); 

PROC IMPORT OUT= work.&ds. DATAFILE= "S:\fiscal\AFR\Submitted 2013 Charter template 

AFRs\&file_name."  

DBMS=EXCEL REPLACE; RANGE="&tab"; GETNAMES=YES; MIXED=NO; SCANTEXT=YES; 

USEDATE=YES; SCANTIME=YES; 

RUN; 

 

/* renaming variables F_1 - F_? so help get CDNum from cell B1 */ 

title1; 

proc sql; 

select max(varnum) into :num_vars 

FROM DICTIONARY.COLUMNS 

WHERE UPCASE(LIBNAME)="WORK" AND UPCASE(memname)="&ds."; 

select distinct(name) into :var1-:var%TRIM(%LEFT(&num_vars)) 

from dictionary.columns 

WHERE UPCASE(LIBNAME)="WORK" AND UPCASE(memname)="&ds." 

order by varnum; 

quit; 

proc datasets library=WORK; 

modify &ds.; 

rename 

%do i=1 %to &num_vars; 

&&var&i=F_&i. 

%end; 

; 

quit; run; 

 

/* getting CDNum from cell B1 */ 

%if &ds. NE COVER %then %do; data &ds.; length CDNum $6; set &ds.; 

CDNum=compress(substr(vlabel(F_2),1,7),"-"); run; %end; 

%mend import_data; 

 

%import_data(Cover$,COVER); 

%import_data(Charter Holder$,CHARTER_HOLDER); 

%import_data(SofFP$,SOFFP); 

%import_data(SofA$,SOFA); 

%import_data(SofExp$,SOFEXP); 

%import_data(SofCF$,SOFCF); 

%import_data(Budget Comparison$,BUDGET_COMPARISON); 

 

/* getting data elements to load into the MS Access DB */ 

data work.CDNum(keep=CDNum); length CDNum $6; set work.Cover(where=(F_1 contains "Charter")); 

CDNum=compress(substr(F_3,1,7),"-"); run; 

 

title1 "CDN"; 

proc sql /*noprint*/; 

select '"'||CDNum||'"' into: CDN  

from work.CDNum; 

quit; 

title1; 

 

data work.COH(keep=CDNum COH1 COH2); set work.SofFP;  

if F_2 = "Cash and Cash Equivalents" then COH1=input(F_3,dollar32.); else if F_2 = "Restricted Cash & Cash 

Equivalents" then COH2=input(F_3,dollar32.); else delete; run;  

proc summary nway data=work.COH; class CDNum; var COH1 COH2; output out=work.COH(drop=_type_ 

_freq_) sum=; run; 

data work.COH(keep=CDNum COH); set work.COH; COH=sum(COH1,COH2); run; 

 

%macro data(var,acct,ds,col); 

data work.&var.(keep=CDNum &var.); set work.&ds.(where=(F_2 = &acct.)); &var.=input(&col.,dollar32.); if 

&var.=. then &var.=0; run; 

%mend data; 

 

%data(LR,"Local revenue",Budget_comparison,F_8); 

%data(SR,"State revenue",Budget_comparison,F_8); 

%data(FR,"Federal revenue",Budget_comparison,F_8); 

%data(TR,"Total Revenue",Budget_comparison,F_8); 

%data(F11,"11  Instruction",Budget_comparison,F_8); 

%data(F12,"12  Instructional Resources & Media Services",Budget_comparison,F_8); 

%data(F13,"13  Curriculum & Instructional Staff Development",Budget_comparison,F_8); 

%data(F21,"21  Instructional Leadership",Budget_comparison,F_8); 

%data(F23,"23  School Leadership",Budget_comparison,F_8); 

%data(F31,"31  Guidance, Counseling & Evaluation Services",Budget_comparison,F_8); 

 

 

%data(F32,"Function 32 Expenses",Budget_comparison,F_8); 

 

 

%data(F33,"33  Health Services",Budget_comparison,F_8); 

%data(F34,"34  Transportation",Budget_comparison,F_8); 

%data(F35,"35  Food Service",Budget_comparison,F_8); 

%data(F36,"36  Extracurricular",Budget_comparison,F_8); 

%data(F41,"41  General Administration",Budget_comparison,F_8); 

%data(F51,"51  Plant Maintenance",Budget_comparison,F_8); 

%data(F52,"52  Security & Monitoring Services",Budget_comparison,F_8); 

%data(F53,"53  Data Processing Services",Budget_comparison,F_8); 

%data(F61,"61  Community Services",Budget_comparison,F_8); 

%data(F71,"71  Debt Service",Budget_comparison,F_8); 

%data(F81,"81  Fundraising",Budget_comparison,F_8); 

%data(TE,"Total Expenses",Budget_comparison,F_8); 

%data(CNA,"CHANGE IN NET ASSETS",Budget_comparison,F_8); 

 

%data(CHCOH,"Charter Holder Cash on Hand",Charter_holder,F_3); 

%data(CHFR,"Charter Holder Federal Revenue",Charter_holder,F_3); 

%data(CHFE,"Charter Holder Federal Expenditures",Charter_holder,F_3); 

 

%data(URI,"Total Revenue and Other Revenue",SofA,F_4); 

%data(CUNA,"INCREASE (DECREASES) IN NET ASSETS",SofA,F_4); 

 

%data(DE,"Depreciation",SofCF,F_4); 

 

%data(O6100,"6100  Payroll Costs",SofExp,F_3); 

%data(O6200,"6200  Professional & Contracted Services",SofExp,F_3); 

%data(O6300,"6300  Supplies & Materials",SofExp,F_3); 

%data(O6400,"6400  Other Operating Costs",SofExp,F_3); 

%data(O6500,"6500  Debt Service",SofExp,F_3); 

%data(TOE,"Total Expenses",SofExp,F_3); 

 

%data(COHPR,"Restricted Cash & Cash Equivalents",SofFP,F_3); 

%data(REC,"Accounts Receivable",SofFP,F_3); 

%data(INV,"Investments",SofFP,F_3); 

%data(PREPD,"Prepaid Expenses",SofFP,F_3); 

%data(CA,"Total Current Assets",SofFP,F_3); 

%data(LND,"Land",SofFP,F_3); 

%data(ADEP,"Accumulated Depreciation",SofFP,F_3); 

%data(PE,"Total Property and Equipment (Net)",SofFP,F_3); 

%data(OA,"Total Other Assets",SofFP,F_3); 

%data(TA,"Total Assets",SofFP,F_3); 

%data(AP,"Accounts Payable",SofFP,F_3); 

%data(NPC,"Notes Payable - Current Maturities",SofFP,F_3); 

%data(OL,"Other Current Liabilities",SofFP,F_3); 

%data(CL,"Total Current Liabilities",SofFP,F_3); 

%data(NPL,"Notes Payable - Long-term",SofFP,F_3); 

%data(TL,"Total Liabilities",SofFP,F_3); 

%data(UNA,"Unrestricted",SofFP,F_3); 

%data(TRNA,"Temporarily Restricted",SofFP,F_3); 

%data(PRNA,"Permanently Restricted",SofFP,F_3); 

%data(TNA,"Total Net Assets",SofFP,F_3); 

 

%macro missing_data(var,CDNum_missing_data); 

%if &CDN. = &CDNum_missing_data. %then %do; 

data work.&var.(keep=CDNum &var.); CDNum=&CDNum_missing_data.; &var.=0; run; 

%end; 

%mend missing_data; 

 

%missing_data(F32,"015820"); 

%missing_data(OtherGainsLosses,"015820"); 

%missing_data(PriorPeriodAdjustments,"015820"); 

%missing_data(O7900,"015820"); 

%missing_data(O8900,"015820"); 

 

%missing_data(F32,"003801"); 

%missing_data(OtherGainsLosses,"003801"); 

%missing_data(PriorPeriodAdjustments,"003801"); 

%missing_data(O7900,"003801"); 

%missing_data(O8900,"003801"); 

 

data work.charter_AFR_&file_number.;  

merge CDNum CA COH COHPR CHCOH OA TA CL OL TL UNA TRNA PRNA TNA INV REC PREPD AP 

NPC NPL DE CUNA URI CHFR CHFE LND PE ADEP LR SR FR TR F11 F12 F13 F21 F23 F31 F32 F33 F34 

F35 F36 F41 F51 F52  

F53 F61 F71 F81 TE OtherGainsLosses PriorPeriodAdjustments CNA O6100 O6200 O6300 O6400 O6500 O7900 

O8900 TOE; 

by CDNum; 

run; 

 

proc append base=work.charter_AFR data=work.charter_AFR_&file_number.; 

run; 

 

%mend file_info; 

%file_info(SASIC Charter AFR Template FYend 2013 v1 1(1).xlsx,_1); 

%file_info(Charter AFR Template FYend 2013 v1 2.xlsx,_2); 

 

%macro export(data); 

 PROC EXPORT DATA= WORK.&data 

 OUTFILE= "S:\fiscal\AFR\Charter AFR data\&data. &filedate..xlsx" 

 DBMS=EXCEL REPLACE; 

 SHEET="&data."; 

RUN; 

%mend export; 

%export(charter_AFR); 


